

## FlexPDE v7 Trial Script [A]

汎用的な偏微分方程式ソルバである FlexPDE の機能を体験していただくために、ここでは 2 次元熱伝導の問題を例にとってスクリプトを作成、それを評価版で実行してみることになります。

### 1 問題設定

次の図に示されるような正方形と円形からなる領域を解析対象ドメインとして想定することになります。




図 1 解析対象ドメイン

これを良導性の角柱の中に熱伝導率の低い円柱が埋め込まれているときの断面図としてとらえるなら、3次元の物理的問題が考察対象であるということもできるわけです。このとき、次のような条件のもとでの熱的定常状態を想定し、そのときの温度分布等を解析したいものとします。

- 正方形領域 Region 1 は 1 辺の長さが  $2m$  の正方形であるとし、その熱伝導率を  $k = 1$  とする。
- 円形領域 Region 2 の直径は  $1m$  であるとし、その熱伝導率としては  $k = 0.001$  を想定する。
- Region 1 の左右境界は絶縁状態におかれ、熱の流出はないものとする。
- Region 1 の上部境界は  $100^{\circ}C$  に、下部境界は  $0^{\circ}C$  に保たれているものとする。

## 2 支配方程式

今、ドメイン内における温度の関数を  $u(x, y)$  としたとき、定常状態における熱伝導方程式は

$$\frac{\partial^2}{\partial x^2}(ku) + \frac{\partial^2}{\partial y^2}(ku) = 0 \quad (1)$$

のように記述できます。 $k$  は熱伝導率を表すパラメータですが、位置によって異なる値を持つことから、微分演算子の中に置いたままとしている点に注意してください。

## 3 境界条件

Region 1 の上下の境界上では温度  $u$  の値が一定であるため、それぞれ  $u = 100, u = 0$  というディリクレ型境界条件を設定することになります。一方、左右の境界上ではそれを横切る熱の流れが存在しないという意味で、 $\frac{\partial u}{\partial n} = 0$  というノイマン型境界条件を設定します。なおこの微分演算子は境界上における法線方向の微分を意味し、この問題では右上上で  $\frac{\partial u}{\partial x}$  と、左上上では  $\frac{\partial u}{\partial(-x)}$  と等価になります。

## 4 スクリプト

それではこの問題を FlexPDE で解いてみましょう。FlexPDE を起動後、最初に必要となるのはスクリプト言語による問題の記述です。言語とは言っても宣言文の羅列に過ぎないため、プログラミング言語に比べるとはるかに単純なものであり、恐れるには及びません

### (1) スクリプトエディタのオープン

新規にスクリプトを作成する場合には FlexPDE の操作画面上で  をクリックします。格納場所を聞いてくれるので、適当なフォルダを選択の上、ファイル名を指定します。ここでは trial7a という名前を用いることにしますが、この場合、フォルダ内には trial7a.pde というファイルが生成されることになります。

新規作成ということなので白紙のスクリプトが期待されるのですが、実際の画面表示は図 2 に示したようなものとなります。TITLE とか VARIABLES といったセクションごとにどのような入力を行ったら良いかが示されているわけですが（`{ }` 内、あるいは!に続くテキストは注釈であり、緑色で表示されています）ここでは全体を消去し、白紙の状態からスタートすることにします。




図2 新規スクリプト

## (2) TITLE セクション

スクリプト (正式名称は problem descriptor) に対する名称を定義します。この名称は出力されるグラフィックスの先頭行にタイトルとして表示されることとなります。ここでは

```
TITLE 'Heat flow around an Insulating blob'
```

のように入力してみてください。

## (3) VARIABLES セクション

このセクションでは従属変数名称を定義します。方程式 (1) 上では温度を表す  $u$  が従属変数なので

```
VARIABLES
U { the temperature }
```

のように指定します。

## (4) DEFINITIONS セクション

このセクションでは方程式や境界条件中に現れるパラメータを定義します。この問題では熱伝導率を表す  $k$  と Region 2 の半径を表す  $r$  を規定します。なお、ここで設定する  $k$  の値はデフォルトとしての意味を持つこととなります。

## DEFINITIONS

```
k = 1 { default conductivity }
r = 0.5 { blob radius }
```

## (5) EQUATIONS セクション

物理場を規定する偏微分方程式を記述します。(1) 式がそれに該当するわけですが、 $\frac{\partial^2}{\partial x^2}$ ,  $\frac{\partial^2}{\partial y^2}$  という演算子をスクリプト上で表現する際にはそれぞれ `dxx`, `dyy` のような記法を用います\*1。

## EQUATIONS

```
dxx(k*U) + dyd(k*U) = 0
```

## (6) BOUNDARIES セクション

このセクションでは解析対象ドメインの構成、及び境界条件を記述します。図 1 に示した問題の場合、熱伝導率という特性に違いがあるため、Region 1 と Region 2 という 2 つのリージョンの設定が必要となります。Region 1 はドメイン全体に、Region 2 は内部の円形部に対応するわけですが、定義はこの順に行います。その場合、Region 2 は Region 1 の一部と重なる形となるわけですが、重なった部分の物理特性は後に設定されたリージョンのものとなる点に注意してください。

最初に Region 1 を設定してみます。

## BOUNDARIES

## REGION 1

```
START (-1,-1)
VALUE(U) = 0 LINE TO (1,-1)
NATURAL(U) = 0 LINE TO (1,1)
VALUE(U) = 100 LINE TO (-1,1)
NATURAL(U) = 0 LINE TO CLOSE
```

基本的には Start  $(x_1, y_1)$  Line to  $(x_2, y_2)$  ... Line to Close という構文によって矩形閉領域の規定を行っているわけですが、それぞれの線分（セグメント）ごとに境界条件の設定が入り込んでいるため、少々わかりにくいかも知れません。ディリクレ型の境界条件は Value 文で、ノイマン型の境界条件（自然境界条件とも言う）は Natural 文で指定します。

\*1 ベクトル解析の演算子を用いて  $\text{div}(k*\text{grad}(U))$  と表記しても構いません。

次に Region 2 を設定します。

```
REGION 2 { the embedded blob }
k = 0.001
START 'ring' (r,0)
ARC(CENTER=0,0) ANGLE=360 TO CLOSE
```

Region 1 中では  $k$  の値を明示しなかったわけですが、その場合には DEFINITIONS セクション中で指定された値、すなわち  $k = 1$  がデフォルトとして用いられます。これに対し Region 2 中では  $k = 0.001$  と明示しているため、その値が優先されることとなります。この Region 2 は円形の閉領域であるため、Start  $(x_1, y_1)$  Arc(Center =  $x_2, y_2$ ) Angle =  $\theta$  To Close という構文が用いられています。なお、この内部境界(円周)に対しては 'ring' という名称が設定してあります。これは後続の PLOTS セクションにおける参照を企図してのものであります。

#### (7) PLOTS セクション

このセクションではどのようなグラフィックスを得たいかを規定します。

```
PLOTS
  GRID(x,y)
  CONTOUR(U)
  VECTOR(-k*grad(U))
  ELEVATION(U) FROM (0,-1) TO (0,1)
  ELEVATION(U) ON 'ring'
```

- Grid(x,y) というコマンドは FlexPDE によって自動生成された有限要素法のメッシュ構成をプロットするものです。
- Contour(U) というコマンドにより、変数  $u$  に関する等高線図、すなわち等温線図を作成することができます。
- これに対し  $-k \text{grad}(U)$  という演算操作によって熱流を表す勾配ベクトルが生成されるので、そのベクトル場を Vector コマンドによってプロットさせることができます。
- 最後に Elevation コマンドが 2 つ設定されていますが、これは指定された直線/曲線上における関数値をプロットする機能を提供します。この例では共に温度  $u$  の値をプロットさせているわけですが、第 1 の Elevation コマンドでは  $y$  軸上の  $(0, -1)$  から  $(0, 1)$  に至る線分に沿って、第 2 の Elevation コマンドでは円周 'ring' に沿って、温度がどう変化しているかを示すこととなります。

#### (8) END セクション

最後にスクリプトの終端を示す END セクションを配置します。

```
END
```

## 5 実行結果


スクリプトが作成できたところで保存操作を行ってください。File ▷ Save Script と操作することで trial7a.pde というファイルの内容が更新されます。



作成済みのスクリプトを開く場合には  アイコンを使用します。

この状態で  アイコンをクリックすると FlexPDE の実行が行えます。次に示すようなグラフィックスが無事出力されたでしょうか？

(1) GRID(x,y)




FlexPDE によって自動生成された有限要素法メッシュの構成を示すプロットです。下部に表示されている Grid#1 という情報から、最初に設定されたメッシュ構成で所定の演算精度が得られたことがわかります。十分な演算精度が得られなかった場合には、自動的にメッシュの細分化が行われます。




演算精度は SELECT セクションにおける指定で制御できます。

## (2) CONTOUR(U)




等温線が色分けされた形でプロットされています。温度  $u$  の値は上部境界上で 100、下部境界上で 0 (正確には  $10^{-3}$ ) であることが確認できると思います。また左右の境界上ではすべての等温線が垂直に交わっている点にも注意してください。

## (3) VECTOR(-k\*grad(U))




熱の流れを表す流線がベクトルプロットで表現されています。これらの流線と等温線とは直交関係にあります。熱伝導率の小さな円形領域の左端と右端で流れが最大となっていることが読み取れます。

## (4) ELEVATION(U) FROM (0,-1) TO (0,1)



プロット右部の小さな図に示されている  $y$  軸上の 2 点 1, 2 を結ぶ直線上で温度  $u$  がどのように変化しているかを示すプロットです。円形領域の内部では温度がリニアに変化していることが示されています。

## (5) ELEVATION(U) ON 'ring'



円形領域の境界線上で温度  $u$  がどのように変化しているかを示す Elevation プロットです。赤字で示された 1, 2, 3, 4 がどの点に対応するかは右側の補助図面上に示されています。約  $8^\circ$  から  $92^\circ$  の範囲で変化する正弦状の曲線がプロットされています。



FlexPDE をインストールした際生成される Samples フォルダには多数のサンプルスクリプトが収納されています。これらについても評価版で実行できますのでお試しください。

